ESP32 Fire, Smoke & Temperature Alert System

Arduino / ESP32 Source Code

```
#include <Wire.h>
#include <Adafruit_GFX.h>
#include <Adafruit_SSD1306.h>
#include <DHT.h>

// OLED
#define SCREEN_WIDTH 128
#define SCREEN_HEIGHT 64
Adafruit_SSD1306 display(SCREEN_WIDTH, SCREEN_HEIGHT, &Wire, -1);

// Sensors
#define MQ_PIN 34
#define FLAME_DO 27
#define DHTPIN 4
#define DHTTYPE DHT11

DHT dht(DHTPIN, DHTTYPE);

// Actuators
#define RELAY_PIN 23
#define BUZZER_PIN 25

// Thresholds (tune these)
int smokeThreshold = 2000;
float tempThreshold = 33.0;

void setup()
{
  Serial.begin(115200);

  pinMode(FLAME_DO, INPUT);
  pinMode(RELAY_PIN, OUTPUT);
  pinMode(BUZZER_PIN, OUTPUT);

  digitalWrite(RELAY_PIN, LOW);
  digitalWrite(BUZZER_PIN, LOW);

  dht.begin();

  if (!display.begin(SSD1306_SWITCHCAPVCC, 0x3C)) {
    Serial.println("OLED not found");
    while (true);
  }

  display.clearDisplay();
  display.setTextSize(1);
  display.setTextColor(WHITE);
}

void loop()
{
  // ===== READ SENSORS =====
  int smokeValue = analogRead(MQ_PIN);
  int flameState = digitalRead(FLAME_DO); // LOW = flame
  float temperature = dht.readTemperature();

  if (isnan(temperature))
  {
    Serial.println("DHT Error");
    return;
  }

  // ===== PRINT NORMAL VALUES =====
  Serial.print("Smoke: ");
  Serial.print(smokeValue);
  Serial.print(" | Flame: ");
  Serial.print(flameState);
  Serial.print(" | Temp: ");
  Serial.println(temperature);

  // ===== CHECK CONDITIONS =====
  bool smokeAlert = smokeValue > smokeThreshold;
  bool flameAlert = (flameState == LOW);
  bool tempAlert  = temperature > tempThreshold;

  bool anyAlert = smokeAlert || flameAlert || tempAlert;

  // ===== OLED NORMAL MODE =====
```

```
  if (!anyAlert)
  {
    digitalWrite(RELAY_PIN, LOW);
    digitalWrite(BUZZER_PIN, LOW);

    display.clearDisplay();

    display.setCursor(0, 0);
    display.print("Smoke: ");
    display.println(smokeValue);

    display.setCursor(0, 15);
    display.print("Flame: ");
    display.println(flameAlert ? "YES" : "NO");

    display.setCursor(0, 30);
    display.print("Temp: ");
    display.print(temperature);
    display.println(" C");

    display.setCursor(0, 50);
    display.println("SYSTEM SAFE");

    display.display();
  }

  // ===== ALERT MODE =====
  else
  {
    digitalWrite(RELAY_PIN, HIGH);   // MOTOR OFF
    digitalWrite(BUZZER_PIN, HIGH);

    display.clearDisplay();

    display.setCursor(0, 0);
    display.println("!!! ALERT !!!");

    int y = 15;

    if (smokeAlert)
    {
      display.setCursor(0, y);
      display.print("Smoke: ");
      display.println(smokeValue);
      y += 12;
    }

    if (flameAlert)
    {
      display.setCursor(0, y);
      display.println("Flame: DETECTED");
      y += 12;
    }

    if (tempAlert)
    {
      display.setCursor(0, y);
      display.print("Temp: ");
      display.print(temperature);
      display.println(" C");
      y += 12;
    }

    display.setCursor(0, y + 5);
    display.println("MOTOR OFF");

    display.display();

    Serial.println("ALERT TRIGGERED -> MOTOR OFF");
  }

  delay(1000);
}
```